

wxMaxima pour l'Analyse Volume II

Auteur : Zachary Hannan, Solano Community College

Exercices du Chapitre 1 et leur corrigé

Présentation

Le manuel wxMaxima for Calculus Volume II, dont l'auteur est Zachary Hannan, propose l'étude de nombreux domaines de l'analyse à l'aide du logiciel Maxima et de son interface graphique wxMaxima. A l'occasion de la traduction française de ce manuel, la correction des exercices de chaque chapitre est proposée. Ce document reprend les exercices du manuel et leur correction à l'aide de wxMaxima.

La dernière version de ce document, ainsi que les fichiers source pour wxMaxima, sont téléchargeables sur le site <https://maxima-french-doc.fr/>, à la rubrique Documentation.

Enoncé des exercices

1. Utiliser `integrate` afin de trouver une approximation numérique de l'intégrale définie $\int_4^7 \frac{1}{x^2 \cdot \sqrt{x-3}} dx$, puis réaliser un graphique ombré à une échelle adaptée afin de mettre en évidence l'aire que représente cette intégrale.
2. Recommencez le premier exercice en définissant une fonction d'aire $A(x)$ et en évaluant la différence $A(7) - A(4)$. Vous êtes libre de choisir n'importe quel point de départ, tant que cela ne génère pas une intégrale qui soit divergente.
3. Recommencez le premier exercice en calculant une primitive $F(x)$ de $f(x) = \frac{1}{x^2 \cdot \sqrt{x-3}}$ et en évaluant $F(x)|_4^7$.
4. Calculer $\int x^2 \cdot \sqrt{1-x^3} dx$ à l'aide d'un changement de variable en u selon le modèle de l'exemple 1.2.1.
5. A l'aide d'un changement de variable calculez $\int_0^{\frac{\pi}{6}} \sin(3x) \cos^3(3x) dx$ comme dans l'exemple 1.2.2. Terminez le calcul en exprimant les bornes de l'intégrale en fonction de u plutôt que d'exprimer la primitive trouvée en fonction de x .
6. Utilisez `trigreduce` pour appliquer la formule de l'angle double à l'intégrande de $\int \cos^4 x dx$. Calculez ensuite séparément chacune des intégrales résultantes en trouvant directement une primitive. Vérifiez le résultat final à l'aide de la commande `diff` suivie de `trigexpand` et de `trigsimp`.
7. Calculer l'intégrale $\int_{-3}^3 \sqrt{9-x^2} dx$ en effectuant une substitution trigonométrique dans le style de l'exemple 1.2.3. Tracer un graphe avec une zone ombrée et un rapport isométrique en utilisant `dimensions`. Expliquer comment cette intégrale peut être calculée à l'aide d'une formule géométrique simple.
8. Définir une fonction `CTS(x, a, b, c)` pour compléter le carré, étant donnés les coefficients d'un polynôme du second degré de la forme $ax^2 + bx + c$. Utiliser cette fonction pour compléter le carré de $2x^2 - 3x + 7$. Vérifier le résultat à l'aide de `expand`.
9. Utiliser `CTS(x, a, b, c)` pour compléter le carré au dénominateur de $\int \frac{dx}{9x^2 + 12x + 10}$. Effectuer un changement de variable en u pour obtenir la forme $\int \frac{1}{u^2 + 1} du$ (avec éventuellement des constantes devant). Finalement, calculer l'intégrale et transformer le résultat en x . Vérifier à l'aide de `diff`.

10. Terminez l'intégrale de l'Exemple 1.3.3 :

$$\int \frac{5x - \frac{2}{3}}{3x^2 + 2x + 2} dx - \int \frac{5}{3(x+1)} dx$$

La deuxième intégrale peut déjà être calculée par la recherche directe d'une primitive. La première intégrale peut être transformée en deux intégrales :

$$\int \frac{du}{u} + \int \frac{\text{constantes}}{3x^2 + 2x + 2} dx$$

Calculez le premier morceau en observant la forme de l'expression, puis utilisez CTS (x, a, b, c) pour transformer le dénominateur dans le second morceau. Effectuez le changement de variable u approprié afin d'obtenir une intégrale que l'on peut calculer directement avec une primitive évidente. Vérifiez enfin votre résultat à l'aide de `diff`.

11. Intégrer $\int x \cdot \sin^2 x dx$ à l'aide d'une intégration par parties comme dans l'exemple 1.3.1.

12. Lorsque nous appliquons une intégration par parties à une intégrale définie, nous obtenons $\int_a^b u dv =$

$$uv|_a^b - \int_a^b v du. \text{ Appliquer cette formule pour calculer } \int_0^2 x \cdot e^x dx.$$

13. Calculez $\int e^{-\alpha t} \cos \beta t dt$ en utilisant une intégration par parties. Vous obtenez alors une équation avec $\int e^{-\alpha t} \cos \beta t dt$ comme membre gauche et les résultats de la première application de l'intégration par parties au membre droit. Répétez en appliquant une nouvelle fois une intégration par parties. Après ces deux intégrations par parties, vous devriez reconnaître un terme contenant l'intégrale originale situé dans le membre droit. Regroupez les deux termes contenant l'intégrale originale dans le membre de gauche, et résolvez pour trouver la valeur de cette intégrale. Vérifiez votre réponse en utilisant `diff`. Tout au long de vos calculs, utilisez `%alpha` et `%beta`. Vous devrez utiliser `declare` pour que `diff` reconnaisse α et β comme des constantes.

14. Créez une fonction `DOS(x, a, b)` pour factoriser $ax^2 - b$ en deux facteurs linéaires, même si a et b ne sont pas des carrés parfaits. Appliquez `DOS(x, a, b)` au dénominateur de $\int \frac{dx}{2x^2 - 5}$. Effectuez ensuite une décomposition en éléments simples dans le style de l'Exemple 1.3.3 et terminez le calcul de l'intégrale par la recherche directe d'une primitive.

15. Calculez l'intégrale de l'exercice précédent en utilisant une substitution trigonométrique dans le style des Exemples 1.2.3 et 1.2.4. Vérifiez que votre réponse est identique à la précédente.

16. Calculez l'intégrale $\int \frac{1}{x \cdot \sqrt{1-x}} dx$ en commençant par la substitution $u^2 = 1 - x$. Une fois l'intégrale transformée en fonction de u , utilisez `partfrac` pour effectuer une décomposition en éléments simples, puis terminez le calcul des intégrales en u par une recherche directe de primitive. N'oubliez pas d'exprimer votre réponse finale en x .

17. Calculez la deuxième intégrale impropre de l'exemple 1.4.3 : $\lim_{t \rightarrow 3^+} \int_t^4 \frac{1}{(x-3)^2} dx$

18. Calculez $\int_0^\infty \frac{1}{x^2 + a^2} dx$ en effectuant une substitution trigonométrique appropriée et en transformant les bornes d'intégration en fonction de la variable de substitution.

Corrigé des exercices

1 Exercice1

(% i1) $f(x) := 1/(x^2 \sqrt{x-3});$

(%o1) $f(x) := \frac{1}{x^2 \sqrt{x-3}}$


(% i2) `'integrate(f(x),x,4,7)=integrate(f(x),x,4,7);`

(%o2) $\int_4^7 \frac{1}{\sqrt{x-3}x^2} dx = \frac{\operatorname{atan}\left(\frac{2}{\sqrt{3}}\right)}{3^{\frac{3}{2}}} - \frac{\pi}{23^{\frac{5}{2}}} + \frac{1}{84}$

(% i3) `'integrate(f(x),x,4,7)=float(integrate(f(x),x,4,7));`

(%o3) $\int_4^7 \frac{1}{\sqrt{x-3}x^2} dx = 0.07608170382713317$

(% i4) `wxdraw2d(
grid=true,
xaxis=true,
yaxis=true,
xrange=[3.5,7.5],
yrange=[-0.1,0.2],
color=black,
explicit(f(x),x,3.5,7.5),
filled_func=true,
filled_func=f(x),
explicit(0,x,4,7),
filled_func=false
);`



(%t4)

(%o4) On peut utiliser la commande romberg qui calcule numériquement une intégrale

(% i5) `romberg(f(x),x,4,7);`

(%o5) 0.07608170460468669 ou la commande quad_qags qui renvoie [valeur, erreur estimée, nb d'itérations, code sortie 0=succès]

(% i6) `quad_qags(f(x),x,4,7);`

(%o6) [0.07608170382713314 , 6.38339928533304610⁻¹³ , 21 , 0]

2 Exercice 2

(% i7) `A(x):=integrate(f(t),t,5,x);`

(%o7) $A(x) := \int_5^x f(t) dt$

(% i8) `A(7)-A(4);`

(%o8) $\frac{\operatorname{atan}\left(\frac{2}{\sqrt{3}}\right)}{3^{\frac{3}{2}}} - \frac{\pi}{23^{\frac{5}{2}}} + \frac{1}{84}$

3 Exercice 3

(% i9) `Fx:=integrate(f(x),x);`

(Fx) $\frac{\sqrt{x-3}}{3(x-3)+9} + \frac{\operatorname{atan}\left(\frac{\sqrt{x-3}}{\sqrt{3}}\right)}{3^{\frac{3}{2}}}$

(% i10) subst(7,x,Fx)-subst(4,x,Fx);

(%o10) $\frac{\operatorname{atan}\left(\frac{2}{\sqrt{3}}\right)}{3^{\frac{3}{2}}} - \frac{\pi}{23^{\frac{5}{2}}} + \frac{1}{84}$

4 Exercice 4

(% i11) kill(all);

(%o0) done

(% i1) 'integrate(x^2*sqrt(1-x^3),x);

(%o1) $\int x^2 \sqrt{1-x^3} dx$ On pose $u = x^3$ comme changement de variable

(% i2) integrale :x^2*sqrt(1-x^3)*del(x);

(integrande) $x^2 \sqrt{1-x^3} del(x)$

(% i3) solve(diff(u)=diff(x^3),del(x));

(%o3) $\left[del(x) = \frac{del(u)}{3x^2} \right]$

(% i4) subst(rhs(%[1]), del(x),integrande);

(%o4) $\frac{\sqrt{1-x^3} del(u)}{3}$

(% i5) subst(u,x^3,%);

(%o5) $\frac{\sqrt{1-u} del(u)}{3}$ On a exprimé l'intégrande en fonction de u. On intègre en u.

(% i6) integrate(coeff(%[1],del(u)),u);

(%o6) $-\left(\frac{2(1-u)^{\frac{3}{2}}}{9}\right)$ Il reste à exprimer en fonction de x

(% i7) subst(x^3,u,%);

(%o7) $-\left(\frac{2(1-x^3)^{\frac{3}{2}}}{9}\right)$ Vérifions avec un calcul direct de l'intégrale

(% i8) integrate(x^2*sqrt(1-x^3),x);

(%o8) $-\left(\frac{2(1-x^3)^{\frac{3}{2}}}{9}\right)$

On peut définir une procédure pour automatiser le changement de variable.changevarint(nom de la nouvelle variable, expression du changement de variable,expression de l'intégrande, variable de l'intégrale)

```
(% i9) changevarint(nomvar, chgt, integrande, x) := block(
[integrande2, temp, temp1, temp2, temp3, sol, temp2bis],
print( Calcul de l'intégrale , 'integrate(integrande, x), avec un changement de variable ),
if nomvar=x then print( Erreur, la nouvelle variable a le même nom que l'ancienne ! )
else (
if freeof(x, chgt) then print( Erreur, le changement de variable n'est pas exprimé
en fonction de la variable qui est dans l'intégrale ! ) else (
print( Le changement de variable est défini par : , nomvar, = , chgt ),
integrande2 : integrande*del(x),
temp : solve(diff(nomvar)=diff(chgt), del(x)),
print( Expression de la différentielle du changement de variable : , temp[1]),
temp1 : subst(rhs(temp[1]), del(x), integrande2),
temp2 : subst(nomvar, chgt, temp1),
temp2bis : coeff(temp2, del(nomvar)),
if not freeof(x, temp2bis) then print( Erreur, le changement de variable n'aboutit pas !
Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable ! ) else (
print( Calcul de l'intégrale après le changement de variable : , 'integrate(temp2bis, nomvar)),
temp3 : integrate(coeff(temp2, del(nomvar)), nomvar),
print( Résultat de l'intégrale après le changement de variable : , temp3),
sol : subst(chgt, nomvar, temp3),
print( Expression de l'intégrale avec la variable de départ : , sol) )))
)$
```

```
(% i10) changevarint(u, x^3, x^2*sqrt(1-x^3), x)$
```

Calcul de l'intégrale $\int x^2 \sqrt{1-x^3} dx$ avec un changement de variable

Le changement de variable est défini par : $u = x^3$

Expression de la différentielle du changement de variable : $del(x) = \frac{del(u)}{3x^2}$

Calcul de l'intégrale après le changement de variable : $\int \sqrt{1-u} du$

Résultat de l'intégrale après le changement de variable : $-\left(\frac{2(1-u)^{\frac{3}{2}}}{9}\right)$

Expression de l'intégrale avec la variable de départ : $-\left(\frac{2(1-x^3)^{\frac{3}{2}}}{9}\right)$

Il existe une commande native dans Maxima : 1) on définit l'intégrale sous forme inerte 2) on définit le changement de variable sous la forme $f(x, u) = u - x^2$ (pour $u = x^2$) 3) on évalue le résultat

```
(% i13) expr : 'integrate(x * exp(x^2), x);
res : changevar(expr, u - x^2, u, x);
ev(res, integrate);
```

(expr) $\int x e^{x^2} dx$

(res) $\frac{\int x e^u du}{2}$

(%o13) $\frac{x e^u}{2}$

```
(% i14) changevarintb(nomvar, chgt, integrande, x, a, b) := block(
[integrande2, temp, temp1, temp2, temp3, sol, temp2bis, a1, b1, valeur],
print( Calcul de l'intégrale , 'integrate(integrande, x, a, b), avec un changement de variable ),
if nomvar=x then print( Erreur, la nouvelle variable a le même nom que l'ancienne ! )
else (
if freeof(x, chgt) then print( Erreur, le changement de variable n'est pas exprimé
en fonction de la variable qui est dans l'intégrale ! ) else (
print( Le changement de variable est défini par : , nomvar, = , chgt ),
integrande2 : integrande * del(x),
temp : solve(diff(nomvar)=diff(chgt), del(x)),
print( Expression de la différentielle du changement de variable : , temp[1]),
temp1 : subst(rhs(temp[1]), del(x), integrande2),
temp2 : subst(nomvar, chgt, temp1),
temp2bis : coeff(temp2, del(nomvar)),
if not freeof(x, temp2bis) then print( Erreur, le changement de variable n'aboutit pas !
Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable ! ) else (
a1 : subst(a, x, chgt), b1 : subst(b, x, chgt),
print( Calcul de l'intégrale après le changement de variable : , 'integrate(temp2bis, nomvar, a1, b1)),
temp3 : integrate(coeff(temp2, del(nomvar)), nomvar),
print( Résultat de l'intégrale après le changement de variable : [ , temp3, ]_ , a1, ^ , b1),
sol : subst(chgt, nomvar, temp3),
print( Expression de la primitive avec la variable de départ : , sol) )),
valeur : integrate(temp2bis, nomvar, a1, b1),
print( Valeur de l'intégrale : , valeur, en valeur approchée : , float(valeur) )
)$
```

```
(% i15) changevarintb(u, x^3, x^2*sqrt(1-x^3), x, 0, 1/2)$
```

Calcul de l'intégrale $\int_0^{\frac{1}{2}} x^2 \sqrt{1-x^3} dx$ avec un changement de variable

Le changement de variable est défini par : $u = x^3$

Expression de la différentielle du changement de variable : $del(x) = \frac{del(u)}{3x^2}$

Calcul de l'intégrale après le changement de variable : $\int_0^{\frac{1}{8}} \sqrt{1-u} du$

Résultat de l'intégrale après le changement de variable : $\left[-\left(\frac{2(1-u)^{\frac{3}{2}}}{9} \right) \right]_0^{\frac{1}{8}}$

Expression de la primitive avec la variable de départ : $-\left(\frac{2(1-x^3)^{\frac{3}{2}}}{9} \right)$

Valeur de l'intégrale : $\frac{\frac{2}{3} - \frac{7^{\frac{3}{2}}}{32^{\frac{1}{2}}}}{3}$ en valeur approchée : 0.04033609925404458

5 Exercice 5

Le changement de variable évident est $u = \cos(3x)$

```
(% i16) changevarintb(u, cos(3*x), sin(3*x)*cos(3*x)^3, x, 0, %pi/6)$
```

Calcul de l'intégrale $\int_0^{\frac{\pi}{6}} \cos(3x)^3 \sin(3x) dx$ avec un changement de variable

Le changement de variable est défini par : $u = \cos(3x)$

Expression de la différentielle du changement de variable : $del(x) = -\left(\frac{del(u)}{3 \sin(3x)} \right)$

Calcul de l'intégrale après le changement de variable : $\frac{\int_0^1 u^3 du}{3}$

Résultat de l'intégrale après le changement de variable : $\left[-\left(\frac{u^4}{12}\right) \right]_0^1$

Expression de la primitive avec la variable de départ : $-\left(\frac{\cos(3x)^4}{12}\right)$

Valeur de l'intégrale : $\frac{1}{12}$ en valeur approchée : 0.0833333333333333

Les bornes de l'intégrale en u sont donc 0 et 1.

6 Exercice 6

(% i17) f(x):=cos(x)^4;
 (%o17) f(x):=cos(x)^4

(% i18) integ:expand(trigreduce(f(x)));
 (integ) $\frac{\cos(4x)}{8} + \frac{\cos(2x)}{2} + \frac{3}{8}$

(% i19) integ2:args(integ);
 (integ2) $\left[\frac{\cos(4x)}{8}, \frac{\cos(2x)}{2}, \frac{3}{8} \right]$

(% i20) I1:integrate(integ2[1],x);
 (I1) $\frac{\sin(4x)}{32}$

(% i21) I2:integrate(integ2[2],x);
 (I2) $\frac{\sin(2x)}{4}$

(% i22) I3:integrate(integ2[3],x);
 (I3) $\frac{3x}{8}$

(% i23) resultat:I1+I2+I3;
 (resultat) $\frac{\sin(4x)}{32} + \frac{\sin(2x)}{4} + \frac{3x}{8}$

On effectue la vérification demandée avec diff, trigexpand et trigsimp

(% i24) resultat2:diff(resultat,x);
 (resultat2) $\frac{\cos(4x)}{8} + \frac{\cos(2x)}{2} + \frac{3}{8}$

(% i25) trigexpand(resultat2);
 (%o25) $\frac{\sin(x)^4 - 6\cos(x)^2\sin(x)^2 + \cos(x)^4}{8} + \frac{\cos(x)^2 - \sin(x)^2}{2} + \frac{3}{8}$

(% i26) trigsimp(trigexpand(resultat2));
 (%o26) $\cos(x)^4$

7 Exercice 7

(% i27) integrale:sqrt(9-x^2);
 (integrande) $\sqrt{9-x^2}$ On pose $x=3\cos(u)$, ce qui permet d'obtenir l'égalité $\sqrt{9-9\cos(u)^2}=3\sin(u)$

Sur $[-3,3]$, on a bien l'équivalence entre $x=3\cos(u)$ et $u=\arccos(x/3)$

On teste la procédure changement de variable

```
(% i28) changevarint(u,acos(x/3),integrande, x);
```

Calcul de l'intégrale $\int \sqrt{9-x^2} dx$ avec un changement de variable

Le changement de variable est défini par : $u = \arccos\left(\frac{x}{3}\right)$

Expression de la différentielle du changement de variable : $del(x) = -\left(3\sqrt{1-\frac{x^2}{9}} del(u)\right)$

Erreur, le changement de variable n'aboutit pas !

Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable !

```
(%o28) Erreur, le changement de variable n'aboutit pas !
```

Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable !

Echec car pas de simplification automatique de l'égalité trigonométrique fondamentale

On effectue donc la substitution trigonométrique manuellement :

```
(% i29) integrande2 :subst(3*cos(u),x,integrande)*del(x);
```

```
(integrande2)  $\sqrt{9-9\cos(u)^2} del(x)$ 
```

```
(% i30) calculdiff :solve(diff(u)=diff(acos(x/3)),del(x));
```

```
(calculdiff)  $\left[ del(x) = -\left(3\sqrt{1-\frac{x^2}{9}} del(u)\right) \right]$ 
```

```
(% i31) rhs(calculdiff[1]);
```

```
(%o31)  $-\left(3\sqrt{1-\frac{x^2}{9}} del(u)\right)$ 
```

```
(% i32) integrande3 :subst(rhs(calculdiff[1]),del(x),integrande2);
```

```
(integrande3)  $-\left(3\sqrt{9-9\cos(u)^2}\sqrt{1-\frac{x^2}{9}} del(u)\right)$ 
```

```
(% i33) integrande4 :subst(3*cos(u),x,integrande3);
```

```
(integrande4)  $-\left(3\sqrt{9-9\cos(u)^2}\sqrt{1-\cos(u)^2} del(u)\right)$ 
```

```
(% i34) integrande5 :trigsimp(integrande4);
```

```
(integrande5)  $-9\sin(u)^2 del(u)$ 
```

```
(% i35) a1 :subst(-3,x,acos(x/3));
```

```
(a1)  $\pi$ 
```

```
(% i36) b1 :subst(3,x,acos(x/3));
```

```
(b1)  $0$ 
```


```
(% i37) integrate(coeff(integrande5,del(u)),u,a1,b1);
```

```
(%o37)  $\frac{9\pi}{2}$  Vérifions par un calcul direct
```

```
(% i38) integrate(sqrt(9-x^2),x,-3,3);
```

```
(%o38)  $\frac{9\pi}{2}$ 
```

```
(% i39) wxdraw2d(
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[-4,4],
  yrange=[-1,4],
  dimensions = [600, 600],
  color=black,
  explicit(sqrt(9-x^2),x,-3,3),
  filled_func=true,
  filled_func=sqrt(9-x^2),
  explicit(0,x,-3,3),
  filled_func=false
);
```



(%t39)

(%o39) Il suffit de remarquer que la fonction est positive et qu'il s'agit d'un demi-disque de rayon 3

```
(% i40) Aire : %pi*3^2/2;
```

(Aire) $\frac{9\pi}{2}$

8 Exercice 8

```
(% i41) CTS(x, a, b, c) := a*(x + b/(2*a))^2 + c - b^2/(4*a)$
```

```
(% i42) expand(CTS(x,a,b,c));
```

(%o42) $ax^2 + bx + c$

(% i43) CTS(x,2,-3,7);
 (%o43) $2\left(x - \frac{3}{4}\right)^2 + \frac{47}{8}$

(% i44) expand(%);
 (%o44) $2x^2 - 3x + 7$

9 Exercice 9

(% i45) expression1 :CTS(x,9,12,10);
 (expression1) $9\left(x + \frac{2}{3}\right)^2 + 6$ On pose donc $u=3/2*x$ soit $x=2/3*u$

(% i46) expression2 :subst(2/3*u,x,expression1);
 (expression2) $9\left(\frac{2u}{3} + \frac{2}{3}\right)^2 + 6$ On factorise l'expression en la manipulant un peu

(% i47) members :args(expression2);
 (members) $\left[9\left(\frac{2u}{3} + \frac{2}{3}\right)^2, 6\right]$

(% i48) factor(members[1])+members[2];
 (%o48) $4(u+1)^2 + 6$ Il faut donc poser $u+1=\sqrt{6}/2*v$ pour obtenir une forme v^2+1
 soit $v=2/\sqrt{6}(u+1)=2/\sqrt{6}(3/2*x+1)$

(% i49) v :2/sqrt(6)*(3/2*x+1);
 (v) $\frac{2\left(\frac{3x}{2} + 1\right)}{\sqrt{6}}$

(% i50) expand(v);
 (%o50) $\frac{3x}{\sqrt{6}} + \frac{2}{\sqrt{6}}$

(% i51) changevarint(w,v,1/(9*x^2+12*x+10),x);

Calcul de l'intégrale $\int \frac{1}{9x^2 + 12x + 10} dx$ avec un changement de variable

Le changement de variable est défini par : $w = \frac{2\left(\frac{3x}{2} + 1\right)}{\sqrt{6}}$

Expression de la différentielle du changement de variable : $del(x) = \frac{\sqrt{6} del(w)}{3}$

Erreur, le changement de variable n'aboutit pas !

Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable !

(%o51) Erreur, le changement de variable n'aboutit pas !

Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable !

Il faut finir le calcul manuellement car pas de simplification automatique qui entraine ce message d'erreur

(% i52) integrande :1/(9*x^2+12*x+10);
 (integrande) $\frac{1}{9x^2 + 12x + 10}$

On appelle z le changement de variable $z=(3x+2)/\sqrt{6}$ ou $x=(\sqrt{6}*z-2)/3$

(% i53) integrale2 :subst((sqrt(6)*z-2)/3,x,integrande)*del(x);
 (integrande2)
$$\frac{\text{del}(x)}{(\sqrt{6}z-2)^2 + 4(\sqrt{6}z-2) + 10}$$

(% i54) calculdiff :solve(diff(z)=diff((3*x+2)/sqrt(6)),del(x));
 (calculdiff)
$$\left[\text{del}(x) = \frac{\sqrt{6} \text{del}(z)}{3} \right]$$

(% i55) rhs(calculdiff[1]);
 (%o55)
$$\frac{\sqrt{6} \text{del}(z)}{3}$$

(% i56) integrale3 :subst(rhs(calculdiff[1]),del(x),integrande2);
 (integrande3)
$$\frac{\sqrt{6} \text{del}(z)}{3 \left((\sqrt{6}z-2)^2 + 4(\sqrt{6}z-2) + 10 \right)}$$

(% i57) integrale4 :factor(integrande3);
 (integrande4)
$$\frac{\text{del}(z)}{3\sqrt{6}(z^2+1)}$$

(% i58) reponse :integrate(coeff(integrande4,del(z)),z);
 (reponse)
$$\frac{\text{atan}(z)}{3\sqrt{6}}$$

(% i59) solution :subst((3*x+2)/sqrt(6),z,reponse);
 (solution)
$$\frac{\text{atan}\left(\frac{3x+2}{\sqrt{6}}\right)}{3\sqrt{6}}$$

(% i60) diff(solution,x);
 (%o60)
$$\frac{1}{6 \left(\frac{(3x+2)^2}{6} + 1 \right)}$$

(% i61) expand(%);
 (%o61)
$$\frac{1}{9x^2 + 12x + 10}$$

10 Exercice 10

(% i62) f2(x) :=5/(3*(x+1));

(%o62) $f2(x) := \frac{5}{3(x+1)}$ A une constante près, on voit qu'une primitive est de la forme $\log(x+1)$; Soit F2 cette primitive

(% i63) F2(x) :=5/3*log(x+1);
 (%o63) $F2(x) := \frac{5}{3} \log(x+1)$

(% i64) diff(F2(x),x);
 (%o64) $\frac{5}{3(x+1)}$

(% i65) integrale1 : (5*x-2/3)/(3*x^2+2*x+2);

(integrande1) $\frac{5x - \frac{2}{3}}{3x^2 + 2x + 2}$ On transforme le numérateur pour faire apparaître $6x+2$, dérivée du dénominateur

(% i66) '(5/6*(6*x+2)-7/3)=expand(5/6*(6*x+2)-7/3);

(%o66) $\frac{5(6x+2)}{6} - \frac{7}{3} = 5x - \frac{2}{3}$ On peut donc séparer en une somme de deux intégrales

(% i67) 'integrate(5/6*(6*x+2)/(denom(integrande1)),x);

(%o67)
$$\frac{5 \int \frac{6x+2}{3x^2+2x+2} dx}{6}$$

(% i68) 'integrate(-7/3/denom(integrande1),x);

(%o68)
$$-\left(\frac{7 \int \frac{1}{3x^2+2x+2} dx}{3}\right)$$

(% i69) integrande2 : (6*x+2)/(3*x^2+2*x+2);

(integrande2)
$$\frac{6x+2}{3x^2+2x+2}$$

(% i70) changevarint(u,3*x^2+2*x+2,integrande2,x);

Calcul de l'intégrale $\int \frac{6x+2}{3x^2+2x+2} dx$ avec un changement de variable

Le changement de variable est défini par : $u = 3x^2 + 2x + 2$

Expression de la différentielle du changement de variable : $del(x) = \frac{del(u)}{6x+2}$

Calcul de l'intégrale après le changement de variable : $\int \frac{1}{u} du$

Résultat de l'intégrale après le changement de variable : $\log(u)$

Expression de l'intégrale avec la variable de départ : $\log(3x^2 + 2x + 2)$

(%o70)
$$\log(3x^2 + 2x + 2)$$

(% i71) CTS(x,3,2,2);

(%o71)
$$3\left(x + \frac{1}{3}\right)^2 + \frac{5}{3}$$

(% i72) integrande3 : 1/(3*x^2+2*x+2);

(integrande3)
$$\frac{1}{3x^2+2x+2}$$
 Le changement de variable à faire est donc $t = 3/\sqrt{5} * (x+1/3)$

(% i73) changevarint(t,3/sqrt(5)*(x+1/3),integrande3,x);

Calcul de l'intégrale $\int \frac{1}{3x^2+2x+2} dx$ avec un changement de variable

Le changement de variable est défini par : $t = \frac{3(x + \frac{1}{3})}{\sqrt{5}}$

Expression de la différentielle du changement de variable : $del(x) = \frac{\sqrt{5} del(t)}{3}$

Erreur, le changement de variable n'aboutit pas !

Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable !

(%o73) Erreur, le changement de variable n'aboutit pas !

Il n'est pas possible d'exprimer l'intégrande en fonction de la nouvelle variable !

Dans ce cas, effectuons le changement de variable manuellement, sachant que $dx = \sqrt{5}/3 * dt$

(% i74) subst((sqrt(5)*t-1)/3,x,integrande3);

(%o74)
$$\frac{1}{\frac{(\sqrt{5}t-1)^2}{3} + \frac{2(\sqrt{5}t-1)}{3} + 2}$$

(% i75) expand(%);

(%o75)
$$\frac{1}{\frac{5t^2}{3} + \frac{5}{3}}$$

(% i76) factor(%);

(%o76)
$$\frac{3}{5(t^2 + 1)}$$

(% i77) integrale4 : 3/(5*(t^2+1))*sqrt(5)/3;

(integrande4)
$$\frac{1}{\sqrt{5}(t^2 + 1)}$$

(% i78) integrate(integrande4,t);

(%o78)
$$\frac{\operatorname{atan}(t)}{\sqrt{5}}$$

(% i79) subst(3/sqrt(5)*(x+1/3),t,%);

(%o79)
$$\frac{\operatorname{atan}\left(\frac{3(x + \frac{1}{3})}{\sqrt{5}}\right)}{\sqrt{5}}$$

Rassemblons ces résultats

(% i80) primitive(x) := F2(x)+5/6*log(3*x^2+2*x+2)-7/(3*sqrt(5))*atan(3/sqrt(5)*(x+1/3))
;

(%o80)
$$\operatorname{primitive}(x) := -F2(x) + \frac{5}{6} \log(3x^2 + 2x + 2) + \frac{-7}{3\sqrt{5}} \operatorname{atan}\left(\frac{3}{\sqrt{5}}\left(x + \frac{1}{3}\right)\right)$$

On vérifie en dérivant

(% i81) diff(primitive(x),x);

(%o81)
$$-\left(\frac{7}{5\left(\frac{9(x + \frac{1}{3})^2}{5} + 1\right)}\right) + \frac{5(6x + 2)}{6(3x^2 + 2x + 2)} - \frac{5}{3(x + 1)}$$

(% i82) expand(%);

(%o82)
$$\frac{30x}{18x^2 + 12x + 12} + \frac{10}{18x^2 + 12x + 12} - \frac{7}{9x^2 + 6x + 6} - \frac{5}{3x + 3}$$

(% i83) factor(%);

(%o83)
$$\frac{x - 4}{(x + 1)(3x^2 + 2x + 2)}$$

(% i84) expand(denom(%));

(%o84)
$$3x^3 + 5x^2 + 4x + 2$$
 CQFD

11 Exercice 11

(% i85) 'integrate(x*sin(x)^2,x);

(%o85)
$$\int x \sin(x)^2 dx$$

(% i86) u : x;

(u)
$$x$$

(% i87) v : integrate(sin(x)^2,x);

(v)
$$x - \frac{\sin(2x)}{2}$$

(% i88) reponse : u*v-integrate(v,x);

(reponse)
$$\frac{x\left(x - \frac{\sin(2x)}{2}\right)}{2} - \frac{\frac{\cos(2x)}{4} + \frac{x^2}{2}}{2}$$

```
(% i89) expand(reponse);
(%o89)  $-\left(\frac{x \sin(2x)}{4}\right) - \frac{\cos(2x)}{8} + \frac{x^2}{4}$ 
(% i90) verif :diff(reponse,x);
(verif)  $\frac{x(1 - \cos(2x))}{2}$  L'égalité trigonométrique  $\cos(2x)=1-2\sin(x)^2$  permet la vérification finale.
```

12 Exercice 12

```
(% i91) 'integrate(x*exp(x),x,0,2);
(%o91)  $\int_0^2 e^x x \, dx$ 
(% i92) u :x;
(u) x
(% i93) v :integrate(exp(x),x);
(v)  $e^x$ 
(% i94) subst(2,x,u*v)-subst(0,x,u*v)-integrate(v,x,0,2);
(%o94)  $e^2 + 1$  Vérifions par un calcul direct
(% i95) 'integrate(x*exp(x),x,0,2)=integrate(x*exp(x),x,0,2);
(%o95)  $\int_0^2 e^x x \, dx = e^2 + 1$ 
```

13 Exercice 13

```
(% i97) declare(%alpha, constant);declare(%beta,constant);
(%o96) done
(%o97) done
(% i98) integpp :exp(-%alpha*t)*cos(%beta*t);
(integpp)  $e^{-(\alpha t)} \cos(\beta t)$ 
(% i99) u :exp(-%alpha*t);
(u)  $e^{-(\alpha t)}$ 
(% i100) uprime :diff(u,t);
(uprime)  $-\left(\alpha e^{-(\alpha t)}\right)$ 
(% i101) v :integrate(cos(%beta*t),t);
(v)  $\frac{\sin(\beta t)}{\beta}$ 
(% i102) 'integrate(integpp,t)=u*v-'integrate(uprime*v,t);
(%o102)  $\int e^{-(\alpha t)} \cos(\beta t) \, dt = \frac{\alpha \int e^{-(\alpha t)} \sin(\beta t) \, dt}{\beta} + \frac{e^{-(\alpha t)} \sin(\beta t)}{\beta}$ 
```

On intègre par parties l'intégrale du deuxième membre, en enlevant le facteur alpha/beta

(% i103) 'integrate(exp(-%alpha*t)*sin(%beta*t),t);

(%o103) $\int e^{-(\alpha t)} \sin(\beta t) dt$

(% i104) u1 :exp(-%alpha*t);

(u1) $e^{-(\alpha t)}$

(% i105) ulprime :diff(u1,t);

(ulprime) $-\left(\alpha e^{-(\alpha t)}\right)$

(% i106) v1 :integrate(sin(%beta*t),t);

(v1) $-\left(\frac{\cos(\beta t)}{\beta}\right)$

(% i107) 'integrate(exp(-%alpha*t)*sin(%beta*t),t)=u1*v1-'integrate(u1prime*v1,t);

(%o107) $\int e^{-(\alpha t)} \sin(\beta t) dt = -\left(\frac{\alpha \int e^{-(\alpha t)} \cos(\beta t) dt}{\beta}\right) - \frac{e^{-(\alpha t)} \cos(\beta t)}{\beta}$

On appelle Intpp l'intégrale de départ. On obtient l'équation

(% i108) equation :Intpp=u*v+%alpha/%beta*(u1*v1-%alpha/%beta*Intpp);

(equation) $\text{Intpp} = \frac{e^{-(\alpha t)} \sin(\beta t)}{\beta} + \frac{\alpha \left(-\left(\frac{e^{-(\alpha t)} \cos(\beta t)}{\beta}\right) - \frac{\alpha \text{Intpp}}{\beta}\right)}{\beta}$

(% i109) solve(equation,Intpp);

(%o109) $\left[\text{Intpp} = \frac{e^{-(\alpha t)} (\beta \sin(\beta t) - \alpha \cos(\beta t))}{\beta^2 + \alpha^2}\right]$ Calculons directement avec maxima l'intégrale

(% i110) integrate(integpp,t);

(%o110) $\frac{e^{-(\alpha t)} (\beta \sin(\beta t) - \alpha \cos(\beta t))}{\beta^2 + \alpha^2}$

14 Exercice 14

(% i111) kill(all);

(%o0) done On suppose que a et b sont positifs (ce qui est le cas dans la suite de l'exercice)

(% i1) DOS(x,a,b) :=(sqrt(a)*x-sqrt(b))*(sqrt(a)*x+sqrt(b));

(%o1) $DOS(x, a, b) := (\sqrt{a}x - \sqrt{b}) (\sqrt{a}x + \sqrt{b})$

(% i2) expand(DOS(x,a,b));

(%o2) $ax^2 - b$

(% i3) factdenom :DOS(x,2,5);

(factdenom) $(\sqrt{2}x - \sqrt{5}) (\sqrt{2}x + \sqrt{5})$

(% i4) factdenom2 :args(factdenom);

(factdenom2) $[\sqrt{2}x - \sqrt{5}, \sqrt{2}x + \sqrt{5}]$

(% i5) equation :1/(2*x^2-5)=a/factdenom2[1]+b/factdenom2[2];

(equation) $\frac{1}{2x^2 - 5} = \frac{b}{\sqrt{2}x + \sqrt{5}} + \frac{a}{\sqrt{2}x - \sqrt{5}}$

(% i6) $1/(2*x^2-5)=\text{expand}(\text{factor}(\text{rhs}(\text{equation})))$;

(%o6) $\frac{1}{2x^2-5} = \frac{\sqrt{2}bx}{2x^2-5} + \frac{\sqrt{2}ax}{2x^2-5} - \frac{\sqrt{5}b}{2x^2-5} + \frac{\sqrt{5}a}{2x^2-5}$ Cette égalité donne les deux équations

(% i7) equ1 :0=b+a;

(equ1) $0 = b + a$

(% i8) equ2 :sqrt(5)*a-sqrt(5)*b=1 ;

(equ2) $\sqrt{5}a - \sqrt{5}b = 1$

(% i9) solve([equ1,equ2],[a,b]);

(%o9) $\left[\left[a = \frac{1}{2\sqrt{5}}, b = -\left(\frac{1}{2\sqrt{5}} \right) \right] \right]$ On en déduit alors

(% i10) 'integrate(1/(2*x^2-5),x)=1/(2*sqrt(5))*integrate(1/factdenom2[1],x)-1/(2*sqrt(5))*integrate(1/factdenom2[2],

(%o10) $\int \frac{1}{2x^2-5} dx = \frac{\int \frac{1}{\sqrt{2}x-\sqrt{5}} dx}{2\sqrt{5}} - \frac{\int \frac{1}{\sqrt{2}x+\sqrt{5}} dx}{2\sqrt{5}}$

Une primitive de chaque intégrale de droite est évidente, on trouve

$\log(\sqrt{2}x-\sqrt{5})/\sqrt{2}$ et $\log(\sqrt{2}x+\sqrt{5})/\sqrt{2}$

ce qui donne comme primitive

(% i11) primitive :1/(2*sqrt(2)*sqrt(5))*log(sqrt(2)*x-sqrt(5))-1/(2*sqrt(2)*sqrt(5))*log(sqrt(2)*x+sqrt(5));

(primitive) $\frac{\log(\sqrt{2}x-\sqrt{5})}{2^{\frac{3}{2}}\sqrt{5}} - \frac{\log(\sqrt{2}x+\sqrt{5})}{2^{\frac{3}{2}}\sqrt{5}}$ On vérifie en dérivant le résultat

(% i12) expand(factor(diff(primitive,x)));

(%o12) $\frac{1}{2x^2-5}$

15 Exercice 15

(% i13) integrande3 :1/(2*x^2-5);

(integrande3) $\frac{1}{2x^2-5}$

(% i14) 5*((sqrt(2/5)*x)^2-1);

(%o14) $5\left(\frac{2x^2}{5}-1\right)$ On pose $x=\sqrt{5/2}/\cos(t)$ comme changement de variable

(% i15) integrande :integrande3*del(x);

(integrande) $\frac{del(x)}{2x^2-5}$

(% i16) eqprov :solve(diff(x)=diff(sqrt(5/2)/cos(t)),del(t));

(eqprov) $\left[del(t) = \frac{\sqrt{2}\cos(t)^2 del(x)}{\sqrt{5}\sin(t)} \right]$

(% i17) diffprov :solve(eqprov[1],del(x));

(diffprov) $\left[del(x) = \frac{\sqrt{5}\sin(t) del(t)}{\sqrt{2}\cos(t)^2} \right]$

(% i18) rhs(diffprov[1]);

(%o18) $\frac{\sqrt{5}\sin(t) del(t)}{\sqrt{2}\cos(t)^2}$

```
(% i19) temp :subst(rhs(diffprov[1]), del(x),integrande);
(temp) 
$$\frac{\sqrt{5} \sin(t) \operatorname{del}(t)}{\sqrt{2} \cos(t)^2 (2x^2 - 5)}$$

(% i20) temp2 :subst(sqrt(5/2)/cos(t),x,temp);
(temp2) 
$$\frac{\sqrt{5} \sin(t) \operatorname{del}(t)}{\sqrt{2} \left( \frac{5}{\cos(t)^2} - 5 \right) \cos(t)^2}$$

(% i21) temp3 :expand(denom(temp2));
(temp3) 
$$5\sqrt{2} - 5\sqrt{2} \cos(t)^2$$

(% i22) temp4 :subst(1-sin(t)^2,cos(t)^2,temp3);
(temp4) 
$$5\sqrt{2} - 5\sqrt{2} (1 - \sin(t)^2)$$

(% i23) temp5 :expand(temp4);
(temp5) 
$$5\sqrt{2} \sin(t)^2$$

(% i24) temp6 :num(temp2)/temp5;
(temp6) 
$$\frac{\operatorname{del}(t)}{\sqrt{2} \sqrt{5} \sin(t)}$$

(% i25) temp7 :integrate(coeff(temp6,del(t)),t);
(temp7) 
$$\frac{\frac{\log(\cos(t)-1)}{2} - \frac{\log(\cos(t)+1)}{2}}{\sqrt{2} \sqrt{5}}$$

(% i26) solution :subst(sqrt(5/2)/x,cos(t),temp7);
(solution) 
$$\frac{\frac{\log\left(\frac{\sqrt{5}}{\sqrt{2x}} - 1\right)}{2} - \frac{\log\left(\frac{\sqrt{5}}{\sqrt{2x}} + 1\right)}{2}}{\sqrt{2} \sqrt{5}}$$
 Il reste à montrer que l'on retrouve bien l'expression de la primitive précédente
(% i27) primitive;
(%o27) 
$$\frac{\log(\sqrt{2x} - \sqrt{5})}{2^{\frac{3}{2}} \sqrt{5}} - \frac{\log(\sqrt{2x} + \sqrt{5})}{2^{\frac{3}{2}} \sqrt{5}}$$
 On transforme la solution et on trouve les mêmes expressions (à une constante près)
(% i28) ratsimp(radcan(solution));
(%o28) 
$$- \left( \frac{\log(\sqrt{2x} + \sqrt{5}) - \log(\sqrt{5} - \sqrt{2x})}{2^{\frac{3}{2}} \sqrt{5}} \right)$$

(% i29) expand(%);
(%o29) 
$$\frac{\log(\sqrt{5} - \sqrt{2x})}{2^{\frac{3}{2}} \sqrt{5}} - \frac{\log(\sqrt{2x} + \sqrt{5})}{2^{\frac{3}{2}} \sqrt{5}}$$

```

16 Exercice 16

```
(% i30) integrande :1/(x*sqrt(1-x));
(integrande) 
$$\frac{1}{\sqrt{1-xx}}$$
 On pose  $u^2 = 1 - x$  comme changement de variable
(% i31) integrande1 :integrande*del(x);
(integrande1) 
$$\frac{\operatorname{del}(x)}{\sqrt{1-xx}}$$

(% i32) solve(diff(u^2)=diff(1-x),del(x));
(%o32) 
$$[\operatorname{del}(x) = -(2u \operatorname{del}(u))]$$

```

```
(% i33) integrale2 :subst(rhs(%[1]), del(x),integrande1);
(integrande2)  
$$-\left(\frac{2u \operatorname{del}(u)}{\sqrt{1-xx}}\right)$$


(% i34) assume(u>0);
(%o34)   $[u > 0]$ 

(% i35) integrale3 :subst(1-u^2,x,integrande2);
(integrande3)  
$$-\left(\frac{2 \operatorname{del}(u)}{1-u^2}\right)$$


(% i36) integrale4 :coeff(integrande3,del(u));
(integrande4)  
$$-\left(\frac{2}{1-u^2}\right)$$


(% i37) partfrac(integrande4,u);
(%o37)   $\frac{1}{u-1} - \frac{1}{u+1}$  On obtient facilement une primitive F(u)

(% i38) F(u) :=log(u-1)-log(u+1);
(%o38)   $F(u) := \log(u-1) - \log(u+1)$ 

(% i39) solution :subst(sqrt(1-x),u,F(u));
(solution)   $\log(\sqrt{1-x}-1) - \log(\sqrt{1-x}+1)$  On vérifie en dérivant le résultat

(% i40) expand(factor(diff(solution,x)));
(%o40)   $\frac{1}{\sqrt{1-xx}}$ 
```

17 Exercice 17

```
(% i41) assume(t-4>0);
(%o41)   $[t > 4]$ 

(% i42) integrate(1/(x-3)^2,x,t,4);
(%o42)   $\frac{1}{t-3} - 1$ 

(% i43) limit(%, t, 3, plus);
(%o43)   $\infty$  Cette intégrale est donc divergente
```

18 Exercice 18

```
(% i44) declare(a, constant);
(%o44)  done

(% i45) integrale :1/(x^2+a^2);
(integrande)   $\frac{1}{x^2+a^2}$  On pose x=atan(u)

(% i46) integrale1 :integrande*del(x);
(integrande1)   $\frac{\operatorname{del}(x)}{x^2+a^2}$ 

(% i47) solve(diff(a*tan(u))=diff(x),del(x));
(%o47)   $\left[\operatorname{del}(x) = a \sec(u)^2 \operatorname{del}(u)\right]$ 
```

```
(% i48)  integrande2 :subst(rhs(%[1]), del(x),integrande1);
(integrande2)  
$$\frac{a \sec(u)^2 \operatorname{del}(u)}{x^2 + a^2}$$


(% i49)  integrande3 :subst(a*tan(u),x,integrande2);
(integrande3)  
$$\frac{a \sec(u)^2 \operatorname{del}(u)}{a^2 \tan(u)^2 + a^2}$$


(% i50)  integrande4 :coeff(integrande3,del(u));
(integrande4)  
$$\frac{a \sec(u)^2}{a^2 \tan(u)^2 + a^2}$$


(% i51)  integrande5 :trigsimp(integrande4);
(integrande5)  
$$\frac{1}{a}$$


(% i52)  chgt :solve(x=a*tan(u),u);
solve : using arc - trig function to get a solution. Some solutions will be lost.
(chgt)  
$$\left[ u = \operatorname{atan}\left(\frac{x}{a}\right) \right]$$


(% i53)  b1 :subst(0,x,rhs(chgt[1]));
(b1)  0

(% i54)  assume(a>0);
(%o54)   $[0 < a]$ 

(% i55)  b2 :subst(t,x,rhs(chgt[1]));
(b2)  
$$\operatorname{atan}\left(\frac{t}{a}\right)$$


(% i56)  b3 :limit(b2,t,inf);
(b3)  
$$\frac{\pi}{2}$$


(% i57)  integrate(integrande5,u,b1,b3);
(%o57)  
$$\frac{\pi}{2a}$$

```